MATH361-09S1 Tutorial 1 Solutions

March 24-26

1.1 In the lectures it was claimed that the discontinuous function

$$u(x,\ t) = \begin{cases} 0 & x < ct \\ e^{-(x-ct)} & x \geq ct \end{cases}$$

was a solution to the linear convection equation

$$u_t + cu_x = 0.$$

Show that this solution does, indeed, satisfy the integral form of this equation. (*Note:* Only the case $a < ct < b$ need be considered since, for other choices, u is continuous and differentiable on the interval $(a,\ b)$.)

Solution:

```
> soln:=(x,t)->piecewise(x<c*t,0,exp(-(x-c*t)));
```

$$soln := (x, t) \mapsto \begin{cases} 0 & x < c\,t \\ e^{-x + c\,t} & otherwise \end{cases}$$

Next we place the assumptions on a, ct and b.

```
> assume(a<c*t,c*t<b);
```

Now integrate u. Remember ~ is Maple's notation to remind us that there are assumptions on these variables.

```
> int(soln(x,t),x=a..b);
```

$$1 - e^{c\sim t\sim - b\sim}$$

```
> diff(%,t);
```

$$-c\sim e^{c\sim t\sim - b\sim}$$

The flux is

```
> phi:=u->c*u;
```

$$\phi := u \mapsto c\,u$$

and so

```
> phi(soln(a,t))-phi(soln(b,t));
```

$$-c\sim e^{c\sim t\sim - b\sim}$$

Therefore we have our result. We can now remove the assumptions.

```
> a:='a':b:='b':c:='c':t:='t':
```

1.2 Pinchover & Rubinstein 1.2(c)

Solution:

```
> pde:=diff(u(x,y),x$4) + diff(u(x,y),y$4) + 2*diff(u(x,y),x$2,
  y$2);
```

$$pde := \frac{\partial^4}{\partial x^4} u(x, y) + \frac{\partial^4}{\partial y^4} u(x, y) + 2 \frac{\partial^4}{\partial x^2 \partial y^2} u(x, y)$$

```
> soln:=exp(alpha*x+beta*y);
```

$$soln := e^{\alpha x + \beta y}$$

```
> eval(pde,u(x,y)=soln);
```

$$\alpha^4 e^{\alpha x + \beta y} + \beta^4 e^{\alpha x + \beta y} + 2 \alpha^2 \beta^2 e^{\alpha x + \beta y}$$

```
> factor(%);
```

$$e^{\alpha x + \beta y} \left(\alpha^2 + \beta^2\right)^2$$

Therefore $\alpha^2 + \beta^2 = 0$; that is

```
> soln1:=eval(soln,alpha=I*beta);
```

$$soln1 := e^{I \beta x + \beta y}$$

Real solutions are

```
> evalc(soln1);
```

$$e^{\beta y} \cos(\beta x) + I e^{\beta y} \sin(\beta x)$$

that is

```
> [op(1,%),-I*op(2,%)];
```

$$\left[e^{\beta y} \cos(\beta x), e^{\beta y} \sin(\beta x)\right]$$

Check!

```
> simplify(eval(pde,u(x,y)=%[1])),simplify(eval(pde,u(x,y)=%[2]));
```

$$0, 0$$

*1.3 Linear homogeneous PDEs with constant coefficients admit complex solutions of the form

$$u(x,\ t) = Ae^{i(kx-\omega t)},$$

which are called *plane waves*. The real and imaginary parts of this complex solution give real solutions. Here A is the amplitude, k is the wave number and ω is the temporal frequency. When the plane wave form is substituted into a PDE there results a *dispersion relation* of the form $\omega = \omega(k)$ which states how the frequency depends on the wave number. For each of the following PDEs, find the dispersion relation and describe the resulting plane waves by sketching (or use MAPLE) wave profiles at different times.

- $u_t = Du_{xx}$ (heat equation).
- $u_{tt} - c^2\, u_{xx} = 0$ (wave equation).
- $u_t + u_{xxx} = 0$ (linearized Korteweg-de Vries equation - approximates waves on shallow water surfaces).
- $u_t = i\, u_{xx}$ (Schrödinger equation).

Solution:

```
> plane_wave:=u(x,t)=A*exp(I*(k*x-omega*t));
```

$$plane\_wave := u(x,t) = A\,e^{I(kx-\omega t)}$$

```
> eval(diff(u(x,t),t)=D*diff(u(x,t),x$2),plane_wave);
```

$$-I A \omega\, e^{I(kx-\omega t)} = -D A k^2 e^{I(kx-\omega t)}$$

```
> omega=solve(%,omega);
```

$$\omega = -I D k^2$$

is the dispersion relation and the solution is

```
> eval(plane_wave,%);
```

$$u(x,t) = A\,e^{I(kx+I D k^2 t)}$$

```
> expand(%);
```

$$u(x,t) = \frac{A\,e^{Ikx}}{e^{D k^2 t}}$$

thus the amplitude decays exponentially but the wave speed is 0.

```
> plot({seq(eval(Re(rhs(%)),{D=1,A=1,k=1}),t={0,1,2})},x=0..10);
```



Note Re( ) is the real part.

```
> eval(diff(u(x,t),t$2)-c^2*diff(u(x,t),x$2)=0,plane_wave);
```

$$-A\omega^2 e^{I(kx-\omega t)} + c^2 A k^2 e^{I(kx-\omega t)} = 0$$

```
> omega=solve(%,omega);
```

$$\omega = (c k, -c k)$$

```
> eval(plane_wave,omega=c*k);
```

$$u(x,t) = A\,e^{I(kx-ckt)}$$

Thus the amplitude remains constant. The wave travels with constant speed c.

```
> plot({seq(eval(Re(rhs(%)),{c=1,A=1,k=1}),t={0,1,2})},x=0..10);
```



```
> eval(diff(u(x,t),t)+diff(u(x,t),x$3)=0,plane_wave);
```

$$-I A\omega\, e^{I(kx-\omega t)} - I A k^3 e^{I(kx-\omega t)} = 0$$

```
> omega=solve(%,omega);
```

$$\omega = -k^3$$

```
> eval(plane_wave,%);
```

$$u(x,t) = A\,e^{I(kx+k^3 t)}$$

In this case the speed of the wave depends on wave number (that is inversely of v
Amplitude remains constant.

```
> plot({seq(eval(Re(rhs(%)),{A=1,k=1}),t={0,1,2})},x=0
```



```
> plot({seq(eval(Re(rhs(%%)),{A=1,k=2}),t={0,1,2})},x=
  # in this case the wave speed is 8X that of the abov
```



```
> eval(diff(u(x,t),t)=I*diff(u(x,t),x$2),plane_wave);
```

$$-I A\omega\, e^{I(kx-\omega t)} = -I A k^2 e^{I(kx-\omega t)}$$

```
> omega=solve(%,omega);
```

$$\omega = k^2$$

```
> eval(plane_wave,%);
```

$$u(x,t) = A\,e^{I(kx-k^2 t)}$$

```
> plot({seq(eval(Re(rhs(%)),{A=1,k=1}),t={0,1,2})},x=0
```